

Heterojen Mal

Foundations • The Valuation Engineer

Bert Craytor • 7 Ağustos 2026

Sürüm 1.0.0

DOI: [10.67330/s1scsv65](https://doi.org/10.67330/s1scsv65)

Türkçe sürüm — yapay zekâ desteğiyle hazırlanmış ve editoryal denetimden geçirilmiştir; İngilizce sürüm esas alınır.

Biçimsel tanım. Heterojen mallar, her biri alıcıların ayrı ayrı değerlediği belirgin bir nitelik demeti olduğu ve nitelikleri ayrı ayrı alınıp satılmadığı için birbirinin yerine geçemeyen mallardır.

Sezgisel çerçeve. Homojen mal, herhangi bir biriminin iş göreceği maldır: bir varil WTI ham petrolü, bir kile 2 numara sarı mısır, şebekeden bir kilovatsaat, belirli bir hisse senedinin bir payı. Bu malların ardındaki derecelendirme ölçütleri ya da teknik şartnameler tam da tek tek birimleri birbirinin yerine geçebilir kılmak için vardır; böylece fiyatlar belirli bir birime değil kategoriye uygulanır. Heterojen mal ise belirli birimin önem taşıdığı maldır — *şu* ev, *o* ev değil; *şu* ikinci el otomobil, yanındaki değil.

Bir malı teknik anlamda heterojen kılan iki özellik vardır. Birincisi, alıcılar piyasadaki birimler arasında değişkenlik gösteren, birbirinden ayırt edilebilir birden çok niteliği önemser. İkincisi, bu nitelikler malın üretim teknolojisi tarafından demet hâline getirilmiştir; ayrı ayrı satın alınabilir değildirler. 200 fit kare brüt yaşam alanını tek başına satın alamazsınız. Bir okul bölgesini bir arsa parçasından bağımsız olarak edinemezsiniz. Nitelikler demetlenmiş olarak gelir ve piyasa, niteliklerin kendilerine değil bütün demetlere fiyat biçer.

Gayrimenkul, kanonik heterojen maldır ve bu özelliği alışılmadık ölçüde güçlü biçimde sergiler. Konum nitelikleri tümüyle yinelenemezdir: hiçbir iki parsel aynı konumu işgal edemez, dolayısıyla her taşınmazın onu piyasadaki diğer tüm taşınmazlardan ayıran en az bir niteliği (koordinatları) vardır. Fiziksel, hukuki ve çevresel nitelikler, bir bütün olarak alındığında her taşınmaza özgü olan bu vektöre ek değişkenlik boyutları ekler.

Değerleme uzmanları kavramla nerede karşılaşır. Satış karşılaştırma yaklaşımındaki her düzeltme, konu taşınmaz ile emsallerin heterojen mallar olduğunun örtük bir kabulüdür. Taşınmazlar homojen olsaydı satış fiyatları zaten doğrudan karşılaştırılabilir olurdu; hiçbir tabloya gerek kalmazdı. Tablo, demet bileşimindeki farklara göre düzeltme yaparak farklı demetlerin fiyatlarını uzlaştırma girişimidir. Tablonun ne *yapmadığını* da belirtmek gerekir: bir satış fiyatını hiçbir zaman nitelik değer katkılarına ayırmaz. Geleneksel tabloda değer katkıları için bir sütun yoktur — yalnızca nitelik nitelik, doğrudan kestirilen düzeltmeler vardır.

Maliyet yaklaşımı heterojenlikle, demeti bileşenlerinden — temel, karkas, kaplamalar, yıpranma — yeniden kurarak ve her birini ayrı fiyatlayarak baş eder. Gelir yaklaşımı ise demeti tek bir nakit akışı profiline soyutlayarak baş eder; kapitalizasyon ve getiri oranları demet düzeyindeki heterojenliği piyasadan türetilmiş bir çarpana soğurur. Üç değerlendirme yaklaşımının tamamı, gayrimenkulün heterojen bir mal olduğu temel olgusuna verilen metodolojik yanıtlardır. Bu olguyla yüzleşip yüzleşmemeleri bakımından değil, onu nasıl aştıkları bakımından birbirlerinden ayrılırlar.

Değerleme uzmanları heterojenlikle, emsal seçimi, eşleştirilmiş satış çözümlemesi ya da “elmayla elmayı karşılaştırma” üzerine akıl yürüttükleri her seferinde gayriresmî olarak da karşılaşır. Ola bildiğinçe çok nitelikte benzeşen emsaller bulma içgüdüğü, heterojenliğe verilen bir uygulamacı yanıttır: konu taşınmaz ile emsal arasındaki demet uzaklığını en aza indirin, demetin gözlenmemiş bileşeni satış fiyatları arasındaki farka daha az katkı versin.

Savunulabilirlik açısından neden önemlidir. Gayrimenkulün heterojenliği, değerlendirme metodolojisinin bir disiplin olarak var olmak zorunda olmasının nedenidir. Taşınmazlar birbirinin yerine geçebilir olsaydı piyasa değeri ilan edilmiş fiyatlardan doğrudan gözlenebilirdi; değerlendirme bir sorgulamaya indirgenirdi. Öyle olmadıkları için değer, ilişkili ama özdeş olmayan taşınmazların fiyatlarından çıkarsanmak zorundadır ve her çıkarım adımı, gerekçelendirilmesi gereken metodolojik bir seçimdir.

Heterojenliği biçimsel olarak kabul etmek, USPAP ve Standart Kuralı savunulabilirliğinin dayandığı birçok uygulama sorusunu berraklaştırır. Emsal seçimi, nitelikler uzayında konu taşınmaza, onlardan geri kazanılan örtük fiyatların genellenebileceği kadar yakın birimleri seçmektir. Bir düzeltmenin makullüğü, altındaki örtük fiyatın piyasa kanıtıyla desteklenebilir olup olmadığına bağlıdır. Yaklaşımlar arasındaki uzlaştırma, *bu* taşınmazın heterojenliğini hangi yaklaşımın en inandırıcı biçimde aştığına ilişkin değerlendirme uzmanının kanaatini yansıtır. Bu savunulabilirlik sorularının her biri, altta yatan koşul “heterojen mal” olarak adlandırıldığında daha keskinleşir.

Çözümlü değerlendirme örneği. Aynı Pacifica, Kaliforniya mahallesinde yakın tarihli iki satış:

Nitelik	Emsal A	Emsal B
GLA (sf)	1,650	1,650
Yatak odası / Banyo	3 / 2	3 / 2
Arsa (sf)	9,000	7,200
Manzara	Var	Yok
Satış fiyatı	\$1,425,000	\$1,280,000

Homojen mal çerçevesi altında bunlar aynı taşınmazdır: özdeş GLA, özdeş oda sayısı, aynı mahalle. Heterojen mal çerçevesi altında ise alıcıların ayrı ayrı değerlediği iki nitelikte (arsa ve manzara) farklılaşan, birbirinden ayrı demetlerdir. \$145,000’lık fiyat farkı, bu iki niteliğin *ortak* katkısıdır — tek bir çiftte birbirine dolanmış iki örtük fiyat.

Örtük fiyatları tek tek geri kazanmak daha fazla çift gerektirir; ideal olarak her seferinde tek bir niteliği değiştiren çiftler. Manzaralı / manzarasız ve büyük arsa / küçük arsa bileşimlerini kapsayan üç ya da dört ek satış, örtük fiyatları ayrı ayrı tanımlamaya başlar. Yeterince çok emsalde yeterince değişkenlik varsa hedonik regresyon bu çözme biçimsel olarak gerçekleştirir; değişkenlik sınırlı olduğunda ise değerlendirme uzmanı, güvenilirliği demetlerin nitelikler uzayında birbirine ne kadar yakın olduğuna açıkça bağlı olan, sınırlandırılmış eşleştirilmiş satış akıl yürütmesiyle yetinmek zorundadır.

Örnek, heterojenliğin neden kök koşul olduğunu gösterir: “özdeş” iki ev bile özdeş değildir ve değerlemenin metodolojik aygıtı tam da aralarındaki uzaklığı ele almak için vardır.

Kaynakça.

- Lancaster, K. J. (1966). A New Approach to Consumer Theory. *Journal of Political Economy*, 74(2), 132–157.
- Rosen, S. (1974). Hedonic Prices and Implicit Markets: Product Differentiation in Pure Competition. *Journal of Political Economy*, 82(1), 34–55.

Çapraz başvurular: nitelikler uzayı; hedonik fiyat fonksiyonu; örtük fiyat.

Ek: Prolog biçimselleştirmesi

Aşağıdaki makine tarafından okunabilir yardımcı dosya (001-heterogeneous-good.pl) bu maddeyi derginin Prolog bilgi tabanında biçimselleştirir. Ayrıca bu makalenin yanında göz atılabilir bir dizgi nüshası (galley) olarak yayımlanır.

```
%% 001-heterogeneous-good.pl
%%
%% Prolog companion to Foundations Vol. 1, Issue 1, Article 001
%% Title: Heterogeneous Good
%% Author: Bert Craytor
%% Concept: heterogeneous_good
%% Version: 0.1.0-draft
%%
%% This file is the Prolog galley for the article: it encodes the
%% article's main claims, worked example, and inference rules in a form
%% that other Prolog code in the journal can consult, query, and verify
%% against. Read alongside the entry.tex / entry.pdf prose.

% =====
% DICTIONARY IMPORT
% =====
%
% The journal-wide dictionary is the source of truth for term definitions
% and relation schemas. The use_module below pulls those facts in at
% runtime; the comments that follow reproduce the relevant entries so a
% reader skimming this file in isolation can see what's been imported.

:- use_module('../.../tools/dictionary/dictionary').
:- dictionary_release('dictionary-2026.1').

% --- From the dictionary (loaded via :- use_module above) -----
% term(heterogeneous_good, "Heterogeneous Good", economic_theory,
%     "Heterogeneous goods are goods that are not interchangeable
%     because each is a distinct bundle of characteristics that buyers
%     value separately, and whose characteristics are not separately
%     traded.", published, 'dictionary-2026.1').
%
% term(characteristics_space, "Characteristics Space", economic_theory,
%     "The characteristics space of a class of heterogeneous goods is
%     the vector space whose coordinates index the attributes that
%     buyers value, and in which each good is represented by the
%     vector of its attribute levels.", published, 'dictionary-2026.1').
%
% term(implicit_price, "Implicit Price", economic_theory,
%     "The implicit price of a characteristic is the partial derivative
%     of the hedonic price function with respect to that characteristic,
%     evaluated at a point in the characteristics space; it represents
%     the marginal market price of an incremental unit of the
%     characteristic, holding all other characteristics fixed.",
%     published, 'dictionary-2026.1').
%
% term(hedonic_price_function, ...).
%
% relation(depends_on, 2, "depends on", [foundations_concept, foundations_concept],
%     foundations_methodology, "...", accepted, 'dictionary-2026.1').
% relation(related, 2, ...).
%
% These facts are NOT declared in this file; the use_module above is
% what makes them available at runtime.
% -----
```

```

:- use_terms([ % Foundations concepts (terms/foundations.pl)
               heterogeneous_good, characteristics_space, implicit_price,
               hedonic_price_function,
               % Practice terms (terms/valuation-practice.pl)
               real_estate, homogeneous_good,
               sales_comparison_approach, cost_approach, income_approach,
               comp_selection, paired_sales_analysis, adjustment,
               reconciliation, uspap, standards_rule ]).
:- use_relations([related/2, depends_on/2]).

% =====
% ARTICLE SUBJECT
% =====

article_concept(heterogeneous_good).
article_issue(volume(1), number(1), year(2026)).
article_id('foundations.2026.001').

% =====
% ARTICLE-LOCAL VOCABULARY
% =====
%
% Terms the article needs that are NOT yet in the dictionary, and that
% the editor decided are too narrow to promote. These are field names
% for the worked-example property records - not concepts in the
% journal's ontology.

local_term(gla_sqft).
local_term(bedrooms).
local_term(baths).
local_term(lot_sqft).
local_term(view).

% =====
% FORMAL DEFINITION (restated as Prolog rule)
% =====
%
% From the formal definition: a good is heterogeneous IF buyers care
% about multiple distinguishable attributes that vary across items in
% the market, AND those attributes are bundled by production rather
% than separately traded.

%! heterogeneous(?Good) is nondet.
%
% True when Good satisfies both bundling criteria. The two predicates
% buyers_value_multiple_attributes/1 and attributes_bundled_by_production/1
% must be asserted for specific goods elsewhere; here they characterize
% the formal definition itself.

heterogeneous(Good) :-
    buyers_value_multiple_attributes(Good),
    attributes_bundled_by_production(Good).

% =====
% INTUITIVE FRAMING - CONTRASTS WITH HOMOGENEOUS GOODS
% =====

% Canonical examples of homogeneous goods cited in the article.

```

```

canonical_homogeneous_example(wti_crude).
canonical_homogeneous_example(no_2_yellow_corn).
canonical_homogeneous_example(kilowatt_hour).
canonical_homogeneous_example(common_stock_share).

% Their distinguishing feature: grading or specification standards make
% individual items interchangeable.
interchangeable_via_grading(wti_crude).
interchangeable_via_grading(no_2_yellow_corn).
interchangeable_via_specification(kilowatt_hour).
interchangeable_via_specification(common_stock_share).

% Canonical heterogeneous goods.
canonical_heterogeneous_example(real_estate).
canonical_heterogeneous_example(used_car).
canonical_heterogeneous_example(unique_machinery).

% Why real estate exhibits heterogeneity unusually strongly:
% locational characteristics are perfectly non-replicable.
buyers_value_multiple_attributes(real_estate).
attributes_bundled_by_production(real_estate).
non_replicable_attribute(real_estate, location).

% =====
% APPRAISER ENCOUNTERS
% =====
%
% The three classical valuation approaches are all methodological
% responses to heterogeneity. They differ in HOW they navigate the
% bundle, not in WHETHER they confront it.

valuation_approach(sales_comparison_approach).
valuation_approach(cost_approach).
valuation_approach(income_approach).

responds_to_heterogeneity(sales_comparison_approach,
    "Reconciles the prices of distinct bundles by adjusting for composition differences; the grid
    ↪ records only adjustments, estimated directly, and never decomposes a price into feature value
    ↪ contributions." ).
responds_to_heterogeneity(cost_approach,
    "Reconstructs the bundle from priced components: foundation, framing, finishes, depreciation." ).
    ↪
responds_to_heterogeneity(income_approach,
    "Abstracts the bundle into a single cash-flow profile; cap/yield rates absorb heterogeneity." ).

% Practitioner-level encounters with heterogeneity.
practitioner_response_to_heterogeneity(comp_selection,
    "Minimize bundle distance between subject and comp in characteristics space." ).
practitioner_response_to_heterogeneity(paired_sales_analysis,
    "Isolate an attribute's contribution by holding other attributes constant." ).
practitioner_response_to_heterogeneity(adjustment,
    "Quantify the price impact of one differing attribute at a time." ).

% =====
% DEFENSIBILITY
% =====
%
% Why naming "heterogeneous good" sharpens USPAP defensibility questions.

```

```

defensibility_consequence(comp_selection,
    "Comps must be close enough in characteristics space that recovered implicit prices generalize
    ↪ .").
defensibility_consequence(adjustment,
    "Each adjustment must be supportable from market evidence of the implicit price.").
defensibility_consequence(reconciliation,
    "Choice among approaches reflects which best navigates THIS property's heterogeneity.").

% Root claim: valuation methodology exists as a discipline BECAUSE
% real estate is heterogeneous. If it were homogeneous, value would be
% directly observable from posted prices.
root_claim(
    "Valuation methodology exists as a discipline because real estate is heterogeneous; in the
    ↪ homogeneous case, value would be directly observable.").

% =====
% WORKED EXAMPLE - PACIFICA COMPS
% =====
%
% Two recent sales in the same Pacifica, California neighborhood. The
% example demonstrates that even "identical" houses are not identical:
% the $145,000 price spread is the joint contribution of differences in
% lot size and view, two implicit prices entangled in a single pair.

% The 8-comp Pacifica dataset is shared across entries 001-006 and
% lives in kb/pacifica_comps.pl. Consulting it here brings property/1,
% in_market/2, attribute_value/3, and sale_price/2 into this article's
% namespace.
:- consult('../..kb/pacifica_comps').

% --- Derived facts from the example -----

% The attributes that differ between the two comps.
differing_attribute(comp_a, comp_b, lot_sqft).
differing_attribute(comp_a, comp_b, view).

% The price differential and what it's attributable to.
price_differential(comp_a, comp_b, 145000).
attributed_to(price_differential(comp_a, comp_b, 145000),
    [implicit_price_of(lot_sqft), implicit_price_of(view)]).

% --- Article's argument about identification -----

% A single pair cannot separately identify the two implicit prices -
% they are entangled in the single $145K observation.
joint_identification_only(price_differential(comp_a, comp_b, 145000),
    [lot_sqft, view]).

% Separate identification requires more pairs varying one attribute at
% a time. With sufficient variation, hedonic regression performs the
% disentanglement formally; with limited variation, the appraiser uses
% bracketed paired-sales reasoning, whose reliability depends on how
% close the bundles are in characteristics space.
identification_strategy(few_pairs,
    "Bracketed paired-sales reasoning. Reliability depends on bundle proximity in characteristics
    ↪ space.").
identification_strategy(many_pairs_with_variation,
    "Hedonic regression (see hedonic_regression).").

```

```

% =====
% INFERENCE RULES
% =====

%! comparable(?A, ?B) is nondet.
%
% A and B are comparable when they are distinct properties in the same
% market. This is the operational definition of "comp" used implicitly
% in the article.

comparable(A, B) :-
    property(A),
    property(B),
    A \== B,
    in_market(A, M),
    in_market(B, M).

%! attribute_difference(?A, ?B, ?Attr, ?DA, ?DB) is nondet.
%
% A and B differ on Attr, with DA the value at A and DB the value at B.
% Use this to enumerate the dimensions on which two comps actually vary.

attribute_difference(A, B, Attr, DA, DB) :-
    comparable(A, B),
    attribute_value(A, Attr, DA),
    attribute_value(B, Attr, DB),
    DA \== DB.

%! identifiable_implicit_price(?Attr) is nondet.
%
% An attribute's implicit price is identifiable from the observed comps
% if there exists at least one pair of comparables differing ONLY in
% that attribute. With our two-comp example this fails for lot_sqft AND
% view (both differ in the same pair), which is exactly the article's
% point.

identifiable_implicit_price(Attr) :-
    comparable(A, B),
    findall(X, attribute_difference(A, B, X, _, _), Diffs),
    Diffs == [Attr].

% =====
% CROSS-REFERENCES
% =====
%
% Concepts the article cross-references (corresponding to the
% \foundationsref{} macros in entry.tex).

cross_reference(characteristics_space).
cross_reference(hedonic_price_function).
cross_reference(implicit_price).

```